

**graph twoway mband** — Twoway median-band plots[Description](#)  
[Options](#)[Quick start](#)  
[Remarks and examples](#)[Menu](#)  
[Also see](#)[Syntax](#)

## Description

`twoway mband` calculates cross medians and then graphs the cross medians as a line plot.

## Quick start

Median-band plot of `y` versus `x`

```
twoway mband y x
```

Overlaid on a scatterplot of the observed data

```
twoway scatter y x || mband y x
```

Same as above, but reduce the scatterplot markers to half their normal size

```
twoway scatter y x, msize(*.5) || mband y x
```

Same as above, but specify 20 bands

```
twoway scatter y x, msize(*.5) || mband y x, bands(20)
```

Add the title “My Title”

```
twoway scatter y x || mband y x, title("My Title")
```

Same as above, but suppress the legend

```
twoway scatter y x || mband y x, title("My Title") legend(off)
```

## Menu

Graphics > Twoway graph (scatter, line, etc.)

## Syntax

```
twoway mband yvar xvar [if] [in] [, options]
```

<i>options</i>	Description
<code>bands(#)</code>	number of bands
<code>cline_options</code>	change look of the line
<code>axis_choice_options</code>	associate plot with alternative axis
<code>twoway_options</code>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All options are *rightmost*; see [G-4] **Concept: repeated options**.

## Options

`bands(#)` specifies the number of bands on which the calculation is to be based. The default is  $\max(10, \text{round}(10 \times \log_{10}(N)))$ , where  $N$  is the number of observations.

In a median-band plot, the  $x$  axis is divided into  $\#$  equal-width intervals and then the median of  $y$  and the median of  $x$  are calculated in each interval. It is these cross medians that `mband` graphs as a line plot.

`cline_options` specify how the median-band line is rendered and its appearance; see [G-3] [cline\\_options](#).

`axis_choice_options` associate the plot with a particular  $y$  or  $x$  axis on the graph; see [G-3] [axis\\_choice\\_options](#).

`twoway_options` are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] [twoway\\_options](#).

## Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:


*Typical use*

*Use with by()*

## Typical use

Median bands provide a convenient but crude way to show the tendency in the relationship between  $y$  and  $x$ :

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. scatter mpg weight, msize(*.5) || mband mpg weight
```




The important part of the above is “mband mpg weight”. On the `scatter`, we specified `msize(*.5)` to make the marker symbols half their normal size; see [\[G-4\] size](#).

## Use with by()

`mband` may be used with `by()` (as can all the `tway` plot commands):

```
. scatter mpg weight, ms(oh) ||
  mband mpg weight ||, by(foreign, total row(1))
```



In the above graph, we specified `ms(oh)` so as to use hollow symbols; see [\[G-4\] symbolstyle](#).

## Also see

[G-2] **graph twoway fplit** — Twoway fractional-polynomial prediction plots

[G-2] **graph twoway lfit** — Twoway linear prediction plots

[G-2] **graph twoway line** — Twoway line plots<sup>+</sup>

[G-2] **graph twoway mspline** — Twoway median-spline plots

[G-2] **graph twoway qfit** — Twoway quadratic prediction plots

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).